

MMSFORTH V2.0 WAS WORTH THE WAIT! (Editorial)

MMSFORTH V2.0 has received several man-months of additional work prior to final release. The extra work-load was painful here, and we know the extra wait has been uncomfortable for some of you. But it's worth it. In late June, MMS started shipping V2.0 System Diskettes for the TRS-80 Model I and Model III. Final work on the MMSFORTH USERS MANUAL, upgrades of THE DATAHANDLER and the GAMES and UTILITIES Diskettes continued for several additional weeks, and the V2.0 MMSFORTH System Cassettes are not expected until the end of August.

That eleventh-hour work has provided many additional features: your own small word processor (THE NOTEPAD) with MMSFORTH source code for further modifications, multiple block buffers, optional keyboard typeahead, our own very complete MMSFORTH USERS MANUAL with over 200 pages of text, program listings and analyses and Appendices, and a very smooth system! All early users tell us that the system is a major winner.

So far, no one has reviewed MMSFORTH V2.0 in any magazine. Do not let this deplorable situation continue! If you can appreciate what's there and think other programmers might be equally interested, please help us to get the word out in as many ways as possible. More users equals more support from MMS and also means new faces and ideas in your user group!

V2.0 is just the first of the new arrivals. In this issue we announce FORTHCOM, our new communications software. And we are completing work on several other major products: our powerful FORTHWRITE word processing module and a complete, professional and very fast General Ledger package. MMSFORTH as a system is becoming an unusually complete one, with many of its parts ready for you now!

Not the least of the MMSFORTH System's components is our continuing support. We apologize for necessarily scanty services during these trying months, and assure you that solid MMSFORTH Newsletters are spinning off again (this May-June issue is a late-arriving example!) and that our over-worked staff is tackling the back mail, preparing new sales literature, and in general putting our house back in order after this orgy of productivity!

New readers are reminded that MMSFORTH licensing permits you to subscribe to this Newsletter, but you must pay an additional \$10.00 per year to do so. Send your payment along with your MMSFORTH serial number, name and address, etc. If you're at all serious about Forth, it's worth it! Yes, you still can get a complete set of issues by subscribing for 1980 as well as 1981, and the earlier, pre-79-STANDARD Forth articles still offer a great wealth of information.

-- A. Richard Miller, Editor 4th Class

GET-TOGETHER

Share your questions and answers with a MMSFORTH User Group, or contact MMS for help to start one in your metropolitan area. Here is our present list of contacts for local MMSFORTH User Groups:

CA: Earl Mortensen, 974 Pleasant Hill Road, Redwood City 94061 (415/367-9882).
CA: Ken Nonomura, 416 Duncan Street, Apt. 5, San Francisco 94131 (415/285-5062).
CA: Morris Herman, 503 Rosario Drive, Santa Barbara 93110 (805/964-7144).
CA: Rich Royea, 6456 Lubau, Woodland Hills 91367 (213/704-6859).
FL: Ron Praver, 1525 N.E. 135th Street, Apt. 17, North Miami 33161 (305/893-2467).
LA: Ed Laughery, 1222 Jason Drive, Denham Springs 70726 (504/665-7537).
MA: Jim Gerow, 1630 Worcester Road, Framingham 01701 (617/443-9521 x3562 days, 872-1882 eves.).
MD: Paul van der Eijk, 5480 Wisconsin Avenue #1128, Chevy Chase 20015 (301/656-2772).
MI: Kim Watt, Box 1013, Berkeley 48072 (313/288-9422).
MI: Bob Zwemer, 6408 South Washington, Lansing 48910 (517/393-9287).
NJ: Paul Zucchini, 148 Bertrand Drive, Princeton 08540 (609/452-3585 days, 921-7629 eves.).
TX: Larry Goforth, 10203-J Golden Meadow, Austin 78758 (512/836-0981).

MMSFORTH OUTCLASSES REST ON BENCHMARK TEST

What's a benchmark test worth? No one test is a complete and fair evaluation of a computer, nor is a general one apt to be the proper yardstick for the particular type of performance you desire in your application. But like EPA automobile gas mileage ratings, appropriately selected benchmark programs can provide valuable comparisons and are of real interest to all computer users who seek some real measure of system efficiency.

One of the more generally recognized benchmark routines was offered as "Project Benchmark" in the June 1980 issue of INTERFACE AGE Magazine. This routine for calculating all prime numbers from 1 to 1000 was rewritten in figFORTH by George Young III, and appears on Page 112 of FORTH DIMENSIONS Vol.II, No.4 along with the following table of performance data on a variety of microcomputer systems. Although speed-ups could be effected by different programming, all coding parallels the original to keep the test runs compatible.

CPU/System	Clock (mHz)	Language	Execution Time
Z80/TRS-80 Model 3	2.03	MMSFORTH	2' 28.7"
AM-100	2	Sierra FORTH	2' 53"
6800	.9	figFORTH	4' 13"
8080/NorthStarDOS	1.84	figFORTH	5' 49"
6502/OhioScientific	2	MicrosoftBASIC	8' 25"
AM-100,16-bit word	2	AlphaBASIC	9' 37"
8080/Heath H8	2.2	B.Harbor BASIC	42'
8080/NorthStarDOS	1.84	MicrosoftBASIC	21' 8"
8080/NorthStarDOS	1.84	usoftCompBASIC	8' 41"
8080/NorthStarDOS	1.84	NorthStarBASIC	41' 13"
8080/NorthStarDOS	1.84	C-BASIC V1.01	77'
Z80/SuperBrain	?	C-BASIC	53'
Z80/NorthStarDOS	4	NorthStarBASIC	19'
" , w/float.pt.bd.	4	"	11' 25"
6800	.9	PercomSuperBAS	73'
6800	.9	SWTP V2.3BASIC	81'

With thanks to both magazines, we list the same code here in MMSFORTH V2.0 (79-STANDARD). For the standard test, enter 999 RUN.BENCH . We invite you to compare the speed of Forth and the unusually high speed of MMSFORTH to other systems.

```

0 ( INTERFACE AGE BENCHMARK PROGRAM, Enter w/upper limit-1 * )
1 : BENCH DUP 2 / 1+ SWAP ." STARTING " CR
2   1 DO DUP I 1 ROT
3     2 DO DROP DUP I /MOD
4       DUP 0= IF DROP DROP 1 LEAVE
5         ELSE 1 = IF DROP 1
6           ELSE DUP 0 > IF DROP 1
7             ELSE 0= IF 0 LEAVE
8               THEN
9                 THEN
10                THEN
11              THEN
12            LOOP
13          IF 4 .R ELSE DROP THEN
14        LOOP DROP CR ." FINISHED. " ;
15 : RUN.BENCH 0 0 0 SET-TIME BENCH TIME ;

```

TX: Jim Shepard, 16210 Arbor Downs Drive, Dallas 75248 (214/661-9702).
TX: Dan Healy, 11511 Katy Freeway, Suite 150, Houston 77079 (713/496-4660 days).
WA: Rod Proctor, 13520 N.E. 29th Place, Bellevue 98005 (206/885-4171 days, 883-1923 eves., also on THE SOURCE).
AUSTRALIA: Peter Wragg, 2 Jilba Street, Indooroopilly, Queensland 4068 (07/378-1623, and CL1641 on The Source).
AUSTRALIA: Dave Dartnall, 20 Eldon Street, Dianella, Western Australia 6062 (09/446-8100).
CANADA: Kalman Fejes, 1149D Meadowlands Drive East, Ottawa, Ontario K2E 6J5 (613/225-2443).
ENGLAND: John Newgas, 1 Philip Court, 89 Hornsey Lane, Highgate, London N6 5LN (01/539-7071 days, 348-6518 eves.).
JAPAN: Akira Akutsu, M.D., 2-176 Issha, Meito, Nagoya, 465.
WEST GERMANY: Nigel Head, Birngartenweg 93, 6100-Darmstadt

NOTE: Program trading is one popular facet of these meetings, but NOT commercial programs and WITHOUT MMSFORTH SYSTEMS aboard! Promote legitimate sharing, discourage pirating, and take care not to jeopardize your own MMSFORTH serial number.

FUN & GAMES

A HIDDEN STARBURST IN THE GAME OF LIFE

Just a bit late for the Fourth (Fortb?) of July, here is a surprising Game of Life sequence for your amusement. It sends eight "seat-ejector" gliders spinning to the corners of the screen in perfect match. We first met glider patterns in Volume 1, No. 4 of the MMSFORTH NEWSLETTER.

Use V2.0's high-speed ALIFE routine if you have it, otherwise LIFE will do. Load the diagonal cross demo pattern, press G, and then Interrupt at the 23rd generation. Reverse the pattern and again press G. At Generation 12, the gliders will form and start their spin into space!

TWINKLE

(by David Huntress, Randolph, Mass.)

This is a nice variation on the popular KALEIDOSCOPE graphics program which was suggested by Dave, and which we wrote and printed back in our Volume 1, Issue 3. If you are about to demonstrate V2.0 to your club, etc., try this little program as an attention-getter while the crowd forms!

```

0 ( TWINKLE, by David Huntress ) : TASK ;
1 DIRBLK 13 + 2 LOADS ( Random numbers & Graphics )
2 0 CONSTANT X 0 CONSTANT Y
3
4 : NEWLOC 64 64 RND RND - ' X ! 24 24 RND RND - ' Y ! ;
5 : OK? Y 20 >= IF X 48 < ELSE 1 THEN ;
6 : SET NEWLOC OK? IF Y X ESET 47 Y - X ESET
7 Y 127 X - ESET 47 Y - 127 X - ESET THEN ;
8 : RESET NEWLOC OK? IF Y X ECLR 47 Y - X ECLR
9 Y 127 X - ECLR 47 Y - 127 X - ECLR THEN ;
10 : TITLE 7 24 PTO ." ANNOUNCING "
11 8 24 PTC ." MMSFORTH V2.0! " ;
12 : TWINKLE BEGIN RANDOMIZE PAGE TITLE
13 200 0 DO SET RESET RESET RESET LOOP
14 0 UNTIL ;
15 TWINKLE

```

SIMON GOES FORTH, IN NEW USERS MANUAL

(by Dave Huntress, Randolph, Mass.)

Dave Huntress' new SIMPLE SIMON game isn't so simple, after all - it blinks, beeps, and stretches your memory like the other brand, but instead of four buttons it runs nine. Dave is giving out this nifty Forth program to call your attention to his other fine games on our newly-expanded MMSFORTH Games Diskette. It costs \$39.95, plus \$2.00 shipping/handling. Try them all, you'll like them!

SIMON isn't in this issue, but we thought you'd like to know that it is treated in depth as an example in Chapter 7 of our new MMSFORTH USERS MANUAL. The MANUAL comes free with Version 2.0, and the first eight chapters of the MANUAL are also available to non-users for \$17.50. A solid 85-page introduction for all Forth users, it also includes detailed MMSFORTH (79-STANDARD based) source code and analyses for this fine game AND for the MMSFORTH Checkbook Balancing program. Users get more: 120 pages of Appendices covering old and new aspects of the MMSFORTH System. MMS is very pleased to offer this major new level of documentation, and invites you to get aboard now.

TOWERS OF HANOI, AN EXCURSION IN RECURSION

(by Dave Lindbergh, Worcester, Mass.)

Did you know that a MMSFORTH wordname can call itself? MMSFORTH has sported this feature, called recursion, from the beginning. But now, to conform with 79-STANDARD definitions, it has been modified so you call the word MYSELF instead of the actual wordname.

Recursive techniques are challenging but can pack a bunch of surprises until you have the hang of it. In general, treat them with caution.

We particularly recommend writing of a routine for solution of factorials, as an excellent first exercise in the power and art of recursive programming. Try it now, and for our solution tune in next issue....

To hold you until then, enjoy this recursive puzzle-solving program by Dave Lindbergh (author of OMNITERM and co-author of FORTHCOM). In the classic TOWERS OF HANOI puzzle, you shuffle a set of varying diameter disks from one of three spikes to another. Moving one disk at a time, you must never place a larger diameter disk over a smaller one. Your goal is to transfer them all in the minimum number of moves for any given number of disks. Dave uses the recursive word IN-HANOI to input the next round. Start the action with 6 disks, by entering 6 HANOI.

```

0 ( 07/05/81 TOWERS OF HANOI PUZZLE, by Dave Lindbergh )
1
2 VARIABLE TWO VARIABLE THREE VARIABLE NUM

```

```

3
4 : SAY-MOVE ." Move " NUM ! THREE ! TWO ! DUP . ." to "
5 TWO @ THREE @ DUP . NUM @ CR ;
6
7 : 1- 4 PICK 4 PICK 4 PICK 4 PICK 1- ;
8
9 : IN-HANOI DUP 1 = IF SAY-MOVE
10 ELSE 1- ROT ROT SWAP ROT MYSELF
11 SAY-MOVE NUM ! ROT SWAP NUM @ 1- MYSELF
12 THEN DROP DROP DROP DROP ;
13
14 : HANOI 1 2 3 4 PICK CR IN-HANOI ;
15

```

PERIPHERAL TALK

PLEASE SHARE YOUR MODS!

On behalf of many of our users, MMS asks you to tell us if you have successfully adapted MMSFORTH to large-capacity disk drives, other storage media, etc. If you are willing to share your patch for the Percom Doubler, Omicron, etc., MMS will attempt to include it in the MMSFORTH Newsletter; if you are charging money for a patch program, let us know about that, too. This is a great way to help your fellow Forthers!

INSIDE TRACK (for advanced users)

DATAHANDLER CUSTOM MODIFICATION TECHNIQUES

If you've followed the last few issues, you should have considerable facility with modifications to THE DATAHANDLER. In this issue, Jill Miller presents an easy modification to THE DATAHANDLER to permit selection matching from the right (trailing) end of the word string. This new feature is a natural for non-ambiguously separating state from city in the MMS-MAIL-1 sample file delivered with THE DATAHANDLER. It also is a fine way to locate and remove those accidental trailing blanks which can mysteriously distort your sorting sequences. Once you start using it, you will discover many other valuable applications as well.

Here are the modifications for THE DATAHANDLER V1.2; if you are still running V1.1 the same changes will apply, but 5 block numbers lower:

Block 29: In the definition of STRING, insert T(railing) between L(eading) and P(erfect).

Block 31: In the definition of \$COMP, insert the following between \$S2 C@ AND and THEN THEN R> :

```
ELSE ?SL 69 = IF DROP I $S2 C@ RIGHT$ $S2 $COMPARE 0= THEN
```

Block 32: In the definition of ?SELECT, insert T between ACASE AL and P=<>RZ", and insert one additional \$COMP immediately after the latter.

These small changes are all that is required. Remember that THE DATAHANDLER may be loaded once from source code and then recompiled (in V1.2, using the new PRECOMPILE word), or may be loaded from source code each time. Either way, be sure that you still have AT LEAST 300 BYTES OF AVAILABLE RAM; if not, sacrifice some other function (LABELS, perhaps) or reduce your maximum file size by 1K!

SOUPING UP THE SEARCH UTILITY

(by Dave Campbell, Franconia, VA)

The SEARCH Utility is one of the most helpful within the MMSFORTH V2.0 package. I was, however, a little concerned with the version as supplied because of its inability to search without editing or without printing and because when printing, I was wasting paper. So here is a major rewrite which does nothing to the original function except pretty it up and insert some reminders for forgetful ones like me. I no longer waste paper, either. Take it as an idea and GO FORTH to generate something to your liking.

The SEARCH function is located in Blocks 75 and 76. I have had a field day with both. First, start with Block 76 by moving the command line from Line 14 to Line 15. Then use the Editor's Page mode to compact SEARCH so it will fit within eight or nine lines. Delete what blank lines you can and move SEARCH so it starts at Line 6 and finishes up in Line 13 or 14, leaving Lines 1 to 5 blank and free for things to come.

Transfer Line 6 of Block 75 to Line 1 of Block 76 and add at the end after SWAP : 1 ED? ! SELECT .

Compact Lines 10 to 15 in Block 75 so they will fit in Lines 2 to 5 of Block 76. After you have compacted them, make the transfer to Lines 2 to 5 of Block 76 and then make the following changes:

```
a) Replace I . CR BEGIN with I 4 .R BEGIN
```

b) Insert PRT so you have 0= IF PRT DUP

Now back to Block 75. Delete Lines 6 to 15 and enter the following:

```
7 : TOP PAGE ." Blocks searched looking for: " $WD $. CR ;
8 : PRT PR? @ IF CR ELSE PRINT CR CR ." Block: " I' . THEN ;
9 : SELECT CR ." Search for" IN$ 31 LEFT$ $WD $!
10 CR ." Print the matches" Y/N DUP PR? ! CR
11 IF ." Edit the matches " Y/N DUP ED? !
12 IF ELSE CR ." PBLK =" SB @ DUP PBLK ! .
13 ." ; write-protect tab removed?" THEN
14 ELSE PRINT CR CR CR ." Searching for: " $WD $. CR CRT
15 THEN CR ENTER TOP ;
```

Last, add VARIABLE PR? at the end of Line 3 in Block 75.

Now FLUSH, saving both Blocks 75 and 76, and try SEARCH to see what further changes you want to make. Have at it!

(Editorial comment: We like this routine, provided you have extra RAM for it. We would worry about using DUP PBLK ! in the SELECT definition, as this automatically opens a possible scanning operation for easy overwriting and leaves PBLK reset afterward. And Dave, SEARCH does have a no-printing, no-editing mode of operation!)

AN OUTPUT DIVERSION UTILITY

(by John Rible of MMS)

"Hello, John, I'm finishing up our new MMSFORTH USERS MANUAL and I want to show a Version 2.0 CATALOG listing with complete wordnames. I think the new NOTEPAD program would make it easy if I could only get CATALOG to dump to the disk instead of to the display. Think you can write a routine for the task?"

"Hmm, Dick, sounds interesting! Let me take a look."

The next morning, John passed this block back over the telephone to MMS. As you now know, it worked fine for our task. It's a great deal more flexible than just that, and may do equally useful tasks for you!

```
0 ( 06/20/81 Divert-Output-to-Memory Utility, by J.Rible )
1 DECIMAL
2 VARIABLE TASK ( Memory pointer ) VARIABLE >MEM
3
4 LABEL OUT-MEM HL PUSH PSW PUSH TASK LHLD 13 CPI =0
5 IF L A MOV 63 ORI A L MOV
6 ELSE A M MOV
7 THEN HL INX TASK SHLD PSW POP HL POP ' CRT 1+ @ JMP
8
9 : OUT>MEM ( addr -> ) DUP TASK ! OUT-MEM 21 MMS 4 + !
10 'S OVER - BLANK >MEM @ EXECUTE CRT ;
11
12 : OUT>BLOCK ( 1stblk# -> ) FLUSH EMPTY-BUFFERS PAD 63 OR 1+>R
13 R@ OUT>MEM R@ SWAP D?# TASK @ R> - 1- 1024 / 1+ * DWTSECS
14 IF ." Disk write-error" THEN ; CR
15 ." Set >MEM with: ' word 2->MEM ! Then: 1stblk# OUT>BLOCK" CR
```

Why do we call the first variable TASK ? Ahh, glad you asked! When first written, this routine started with the conventional : TASK ; and the first variable was named MPTR. But we were using it with CATALOG for five-column listings and the top six entries were from this routine. By combining the top two words into one, we get exactly one line of extra words in the CATALOG listing, easily deleted without needing to shift the other entries!

This utility is a beauty! Here's how we used it to write the CATALOG listing in our USERS MANUAL's Appendix A8-1:

Create and load the above block.
Set >MEM to do a CATALOG, with: ' CATALOG 2->MEM !
Then we did a CATALOG onto relative Block 80 (and following) on Drive 1, with: 80 :1 OUT>BLOCK
Next we reloaded THE NOTEPAD and used it to load these blocks:
80 :1 NOTEPAD
The rest was easily done by editing in the additional information with THE NOTEPAD, using shift-control-U to update the corrected version back to disk, and finally using its page-print command (shift-control-P) to dump from RAM to printer. Voila!

ANOTHER HEX-ALPHA DUMP UTILITY

(by Truman Krumholz, Springfield, Missouri)

Here's the first user-contributed V2.0 item for the Newsletter! Truman sent it in with his enthusiastic comments, just a few days after he received his upgrade to MMSFORTH Version 2.0.

We previously published a Hex-Alpha memory dump utility, ADUMP by Andy Watson, back in our very first Newsletter. It is still a good one, and translates easily to V2.0. It will show more information - a full 256 bytes - per screen in a style many will find easier to read, and it can dump as much continuous code as you request using ADUMP in the same format as the normal DUMP command.

But others will prefer the line by line, Model II- and Model III-TRSDOS-like juxtaposition of Hex and Alpha data in Truman's new version, and we are very impressed with the rich combination of ideas in his tight-packed but very neat single block of source code. MMS does extensive editing to most Newsletter offerings, but this one arrived ready to run and to read! Note the use of recursive code in his final definition, HEX-ALF .

```
0 ( Hex-Alpha Dump -- MMSFORTH V2.0 -- T. Krumholz -- 07/10/81 )
1 PAGE 7 11 PTC ." Hex-Alpha Memory Dump - MMSFORTH V2.0"
2 : TASK ; 9 24 PTC ." T. Krumholz" VARIABLE AD VARIABLE AB HEX
3 : INPUT# ." Hex Starting Address" #IN DUP AD ! AB ! ;
4 : HEXDMP AD @ O <# # # # #> TYPE ." H Hex-> " 10 O DO AD
5 @ I + C@ O <# # # # #> TYPE SPACE LOOP CR ;
6 : CHRDM 7 SPACES ." Asc-> " 10 O DO AD @ I + C@ DUP DUP
7 20 < IF 3 SPACES DROP DROP ELSE 80 > IF 3 SPACES
8 DROP ELSE EMIT 2 SPACES THEN THEN LOOP CR ;
9 : DISPLAY 7 O DO HEXDMP CHRDM AD @ 10 + AD ! LOOP ." <ENTER> "
10 ." = Advance <A> = Start over Another key ="
11 ." End " O 31 PTC ." Offset =" AD @ AB @ - 1- . ;
12 : HEX-ALF INPUT# 0 21 PTC ." - HEX-ALPHA -" BEGIN 1 O PTC
13 DISPLAY KEY DUP 41 = IF 1C EMIT 1E EMIT 1C EMIT
14 ." New " MYSELF THEN OD = IF 0 ELSE 1 THEN UNTIL ;
15 PAGE HEX-ALF DECIMAL PAGE FORGET TASK
```

MMSFORTH QUICKIES

USING MOD TO MEASURE YOUR OUTPUT

If you haven't yet met Forth's Modulo functions, here's a great way to get acquainted! The MOD function is available in larger versions of FORTRAN, BASIC, etc. But Forth introduces it to your little TRS-80 to help it do big jobs. MOD comes in plain vanilla, and in fancy combinations such as /MOD, */MOD, U/MOD, et al. Mod means modulo, or "divided by", and is used to provide a modulus, or remainder. Thus, 78 5 MOD returns 3, and 16210 15360 - 64 /MOD returns the line and column positions for the video display's location 16210 in RAM.

The following definition will output a "ruler" to help you lay out your screen displays or printed reports. Try it out, then analyze its code:

```
: SCALE ( width SCALE -> )
CR ." 0" 1- 10 /MOD 1+ 1
DO ." ----" I 10 MOD 48 + EMIT LOOP DUP
IF 1+ 1
DO I 5 = IF ." +" ELSE ." -" THEN LOOP
ELSE DROP
THEN
;
```

We have Allyn Richardson of West Groton, Mass. to thank for these final words on the theory of Modulo functions:

A lady of eighty named Gertie,
Had a boyfriend of fifty named Bertie.
She told him emphatically
That, viewed mathematically,
"By Modulo fifty, I'm thirty!"

LABEL-WRITING ROUTINE

"Hold on, Jill, I've got an important item to add to that errata sheet for our first V2.0 users!"

"Oh no, Dick, the sheets are back from the printer and were about ready for mailing this afternoon. What's more, the printer will be shut down until Monday morning!"

"Can we handwrite the new info in? It's only two lines..."

"Two lines times nearly one hundred sheets? I suppose so, but what a drag!"

"Not to worry, Jill! Wait ten minutes while Forth, some press-on one-inch labels, and a printer can help!"

```
0 ( 07/17/81 Label-writer, by A.R.Miller ) : TASK ;
1
2 : INIT ( Condensed, for Epson MX-80 ) PRINT 15 EMIT CRT ;
3
4 : MESSAGE ." Insert a new line 1 line ahead of it, reading:"
5 CR ." 2CCONSTANT CONSTANT , ;CODE XCHG ' 2@ 1+ JMP
6 CR ( Room for 3 additional lines, here! )
7 CR
8 CR
9 CR ;
10
11 : PRINT-LABELS ." How many labels" #IN 1+ 1 CR INIT
12 DO I 4 .R PRINT MESSAGE CR CRT LOOP ;
13
14 PRINT-LABELS
15
```

MMSFORTH MODIFICATIONS

PRINTER-DRIVER OPTIONS IN V2.0

One can run one's own printer-driver in MMSFORTH V2.0 - as in our prior versions, just be sure to save all registers between character outputs! But V2.0 also comes with an interesting variety of printer-driver options right aboard. Look them over!

First of all, the standard Printer-Driver routine is now provided in MMSFORTH and Assembler source code for your (or our) further modification.

Second, an Extended Printer-Driver option begins on Block 30. It will give your printer many paper formatting capabilities. Since the printer-driver normally is part of the precompiled Forth which comes up when you boot the system, you will want to modify the block(s) appropriately, then boot the system and use Block 15, the Option Select Block, to load from Blocks 16-39 in your preferred manner.

If your printer's type font can print the TRS-80 graphics character set, swap some parens in Block 29 or 31 to take advantage of this. You probably will wish to adjust Block 59 to screen-print with graphics, as well.

Another optional item on either printer-driver is whether to use the TRS-80 ROM's printer-driver routine at all - swapping of a few parens will totally bypass it, which we find desirable for the Epson printer (it speeds paging operations) but a disaster on some others. You also can inhibit a forced space before CR. Test for best!

HOT PATCHES FOR VERSION 2.0

What, already? We have added a few improvements and filled a few pot-holes. If you are still running a test-site Version 2.X diskette, you should have returned it by now for your complete update to V2.0 - newer patches will not work in V2.X, which has been completely replaced by V2.0. If you forgot, do it now! If your new Version 2.0 diskettes date before 7/17/81, you should have already received a 2-page pink "Modification Advisory" of that date. If we missed you (and if your System License has already been processed), please contact MMS.

Also, early Manuals did not note that THE NOTEPAD requires the Limit marker to be inserted while in Line mode - it then gets recognized by NOTEPAD upon entry into Page mode.

NEW AT MMS

FORTHCOM IS HERE!

Hot on the heels of MMSFORTH Version 2.0, MMS expects to have FORTHCOM in distribution by the first week of August.

FORTHCOM requires 32K, one drive, and MMSFORTH V2.0. It provides the following remote communications capabilities: dumb terminal mode with full ASCII keyboard for those mainframe interfacing jobs, transmit or receive Forth blocks with another FORTHCOM system, and one system can host another (permitting remote operation of that system, even with no one home!). FORTHCOM provides system security features, error checking, and data compression of Forth source code at rates up to 1200 baud. It comes complete with MMSFORTH source code, of course, so you can add your own routines, and it only costs \$39.95.

FORTHCOM was done by the experts - MMSFORTH's Tom Dowling, and Dave Lindbergh of OMNITERM fame. Yes, Dave is another MMSFORTH fan, too! MMS has done its share to encourage you to sample the power of telecommunications in MMSFORTH - are you ready to start?

NEW BOOK: INVITATION TO FORTH, by Harry Katzan, Jr.

New since our prior issue, this hardcover book is an interesting one. It seems to be based on a version of figFORTH which is generally compatible with 79-STANDARD and MMSFORTH. As such, it is the best match yet (except, of course, for our own new MMSFORTH USERS MANUAL). "Invitation to Forth" is strong on introductory information, hardware considerations, and on the basic Forth word set. It might have tackled more advanced words, and should have provided a few solid application examples. But then, it does call itself an invitation, not the whole party!

"Invitation to Forth" is available at MMS, and costs \$17.50 plus \$2.00 shipping/handling.

NEW BOOK: MMSFORTH USERS MANUAL

It comes free with MMSFORTH V2.0, of course, and we're quite pleased with our own new level of Forth documentation. Non-users of our Forth can still buy the first eight chapters for \$17.50 plus shipping/handling. (Please note that the number of pages, printing expense, and selling price are up from the originally advertised \$15.00!) This major elaboration of the microFORTH PRIMER (originally from Forth Inc.) now talks 79-STANDARD and MMSFORTH. It includes about 90 pages and covers basic Forth information, up to and including sample application programs with detailed analyses. The non-user version does NOT include the system-specific information in the fat Appendix sections which come with V2.0.

MORE QUICKIES

SCREEN-SCAN

As you will see, this is a quicky in more ways than one! SCREEN-SCAN is a variation on the theme of the PAINT demonstration in MMSFORTH Newsletter 1:1 and in your new MMSFORTH USERS MANUAL. Let's define PAINT as before:

```
: PAINT 15360 1024 ROT ;
```

Now create the following definition to fill the screen with any keys you press:

```
: FILL-IT BEGIN KEY PAINT 0 UNTIL ;
```

Go ahead, just press any printable keys you like!

How about letting the computer push all the keys for you, like this:

```
: SCREEN-SCAN BEGIN 192 32 DO I PAINT LOOP 0 UNTIL ;
```

Try it out! If we are going too fast for you, rewrite SCREEN-SCAN to include a pause routine just before LOOP.

=====

THE LAST WORD: "Memory crash: When an irresistible FORTH meets the immovable object code."

- Peter Wragg, Brisbane, Australia

(Peter adds, "This may leave ROM for improvement!")

(Editor: Hmm, good puns, but a bit wragged...)

